The object of this column is to enhance our readers' collections of interesting and novel problems in chemical engineering. Problems of the type that can be used to motivate the student by presenting a particular principle in class, or in a new light, or that can be assigned as a novel home problem, are requested, as well as those that are more traditional in nature and that elucidate difficult concepts. Manuscripts should not exceed 14 double-spaced pages and should be accompanied by the originals of any figures or photographs. Please submit them to Professor James O. Wilkes (e-mail: wilkes@umich.edu), Chemical Engineering Department, University of Michigan, Ann Arbor, MI 48109-2136.

# *Computer-Facilitated Mathematical Methods in ChE*
# SIMILARITY SOLUTION

VENKAT R. SUBRAMANIAN
*Tennessee Technological University • Cookeville, TN 38505*

High-performance computers coupled with highly efficient numerical schemes and user-friendly software packages have helped instructors teach numerical solutions and analysis of various nonlinear models more efficiently in the classroom. One of the main objectives of a model is to provide insight about a system of interest. Analytical solutions provide very good physical insight, as they are explicit in the system parameters. Having taught applied math to both senior undergraduate and first-year graduate students for five years, this author feels that students do not appreciate the value of analytical solutions because (1) they wrongly believe numerical methods are best used to solve complex problems with high-speed computers, and (2) they are not comfortable or confident doing the complicated integrals, rigorous algebra, and transformations involved in obtaining analytical solutions. Such solutions, however, can be gained using various computer techniques. For example, computer algebra systems such as Maple,[1] Mathematica,[2] MATLAB,[3] and REDUCE,[4] can be used to perform the tedious algebra, manipulations, complicated integrals, variable transformations, and differentiations, etc., involved in applying mathematical methods.

The goal of this paper is to show how Maple can be used to facilitate similarity transformation techniques for solving chemical engineering problems. The utility of Maple in performing the math, solving the equations, and plotting the results will be demonstrated. For an understanding of the physics in the problems solved, readers are advised to refer to the cited references. For the sake of readers not familiar with Maple, a brief introduction about Maple is given.



***Venkat Subramanian*** *is an assistant professor in the Department of Chemical Engineering at Tennessee Technological University. He received a B.S. degree in chemical and electrochemical engineering from Central Electrochemical Research Institute in India, and his Ph.D. in chemical engineering from the University of South Carolina. His research interests include modeling, control and simulation of electrochemical systems including batteries, fuel cells, hybrids, and multiscale simulation. He is the principal investigator of the Modeling, Analysis, and Process-Control Laboratory for Electrochemical Systems (MAPLE lab, <http://iweb.tntech.edu/vsubramanian>).*

## MAPLE

Maple[1] is a computer-algebra system capable of performing symbolic calculations. Although Maple can be used for performing number crunching or numerical calculations just like FORTRAN, the main advantage of Maple is its symbolic capability and user-friendly graphical interface. In a Maple program, commands are entered after a ">". Maple prints the results if a ";" is used at the end of the statement. This helps in fixing mistakes in the program after a particular step, as the results are shown after every step or command. For brevity, in this paper most of the Maple commands are ended with a colon (:). In general, while Maple is very useful in doing transformations, the user might have to manipulate resulting expressions from a Maple command to obtain the equation in the simplest or desired form. Often, these manipulations can be done in Maple itself by "seeing" the resulting expressions. Hence, first-time users should use a ";" instead of a ":" at the end of each statement to view the results after each command/statement. Many of the mistakes made by students are identified and rectified easily if they replace ":" with ";" in all of the statements. Maple can be used to perform all steps from setting up an equation to analyzing the final plots on the same sheet. All the mathematical steps and manipulations involved can be performed in the same program or file. For clarity between the Maple commands and output, all the text describing the process or Maple commands is given within brackets, "[ ]".

## SIMILARITY TRANSFORMATION FOR PARTIAL DIFFERENTIAL EQUATIONS

Similarity transformation is a powerful technique for treating partial differential equations arising from heat, mass, momentum transfer, or other phenomena, because it reduces the order of the governing differential equation (from partial to ordinary). Depending on the governing equation, boundary conditions, domain, and complexity, the similarity transformation technique might yield a closed-form solution, a series solution, or a numerical solution. One of the major difficulties students encounter is that they find it very difficult to convert the governing equation from the original independent variables to a similarity variable. The following examples illustrate the use of computers and software in teaching/obtaining similarity solutions for various chemical engineering problems.

### *Example 1*
### Diffusion/Heat Transfer in Semi-infinite Domains

Consider the transient heat-conduction problem in a slab.[1,2] The governing equation and initial/boundary conditions are expressed in Eq. (1).

$$\frac{\partial u}{\partial t} = \alpha \frac{\partial^2 u}{\partial x^2}$$
$$u(x,0) = 0 \tag{1}$$
$$u(0,t) = 1 \text{ and } u(\infty,t) = 0$$

where u is the temperature, x is the distance from the surface of the slab, t is the time, and $\alpha$ is the thermal diffusivity. Eq. (1) is solved by using the transformation $\eta = x/\left(2\sqrt{\alpha t}\right)$. The original partial differential equation is converted to an ordinary differential equation in the similarity variable, $\eta$. The boundary conditions for U (u in the similarity variable), are:

$$U(0) = 1$$
$$U(\infty) = 0 \tag{2}$$

The steps involved in the similarity transformation method are illustrated below:

Typically, Maple programs are started with a "restart" command to clear all the variables. Next, the "with(student)" package is called to facilitate variable transformations:

```
>restart: with(student):
>eq:=diff(u(x,t),t)-alpha*diff(u(x,t),x$2);
```

$$eq := \left(\frac{\partial}{\partial t} u(x,t)\right) - \alpha\left(\frac{\partial^2}{\partial x^2} u(x,t)\right)$$

[First, u(x,t) is transformed to U($\eta$(x,t)). Then, the governing equation is converted to the similarity variable:]

```
>eq1:=changevar(u(x,t)=U(eta(x,t)),eq):eq2:=expand
(simplify(subs(eta(x,t)=x/2/(alpha*t)^(1/2),eq1))):
eq2:=expand(eq2*t):eq2:=subs (x=eta*2*(alpha*t)^(1/
2),eq2):eq2:=convert(eq2,diff):
```

[The final form of the governing equation is:]

```
>eq2:=expand(-2*eq2);
```

$$eq2 := \left(\frac{d}{d\eta} U(\eta)\right)\eta + \frac{1}{2}\left(\frac{d^2}{d\eta^2} U(\eta)\right)$$

[The given boundary conditions are used to solve the governing equation:]

```
>bc1:=U(0)=1;
```

$$bc1: = U(0) = 1$$

```
>bc2:=U(infinity)=0;
```

$$bc2: = U(\infty) = 0$$

```
>U:=rhs(dsolve({eq2,bc1,bc2},U(eta))):
>U:=convert(U,erfc);
```

$$U: = \operatorname{erfc}(\eta)$$

```
>u:=subs(eta=x/2/(alpha*t)^(1/2),U);
```

$$u := \operatorname{erfc}\left(\frac{x}{2\sqrt{\alpha t}}\right)$$

[The solution is plotted in Figure 1, which shows how the temperature, u, penetrates to progressively greater distances as the time, t, increases:]

```
>plot3d(subs(alpha=0.001,u),x=1..0,t=500..0,axes=bo
xed,labels=[x, t,"u"],orientation=[-60,60]);
```




*Figure 1. Dimensionless temperature distribution in a semi-infinite domain.*

### Example 2
### Plane Flow Past a Flat Plate—Blasius Equation

The velocity distribution in the boundary layer of a plane laminar flow past a flat plate is given by Eq. (3):

$$\frac{\partial u}{\partial x}+\frac{\partial v}{\partial y}=0$$
$$u\frac{\partial u}{\partial x}+v\frac{\partial u}{\partial y}=\frac{\partial^2 u}{\partial y^2}$$
$$u(0,y)=1 \tag{3}$$
$$u(x,0)=0 \text{ and } u(x,\infty)=1$$
$$v(x,0)=0$$

For this problem, first the velocities, u and v, should be converted to stream functions defined by $u=\partial\psi/\partial y$ and $v=-\partial\psi/\partial x$. The stream function, by default, satisfies the continuity equation (Eq. 1). The second equation yields the governing equation for the stream function, $\psi$. Next, the stream function is expressed as $\psi=\sqrt{x}\,f(\eta)$, where $\eta=y/\sqrt{x}$ is the similarity variable. The boundary conditions for u and v yield the boundary conditions for $\psi$, and finally for $f(\eta)$. Once the function $f(\eta)$ is obtained (numerically), both stream functions and velocity expressions can be expressed in terms of f and $\eta$. The steps involved in this example are more tedious compared to the previous example. All the complicated steps involved can be facilitated using Maple:

```
>restart:with(student):with(plots):
```
Warning, the name changecoords has been redefined

[The governing equation is entered:]

```
>eq:=u(x,y)*diff(u(x,y),x)+v(x,y)*diff(u(x,y),y)- diff(u(x,y),y$2);
```

$$eq := u(x,y)\left(\frac{\partial}{\partial x}u(x,y)\right)+v(x,y)\left(\frac{\partial}{\partial y}u(x,y)\right)-\left(\frac{\partial^2}{\partial y^2}u(x,y)\right)$$

[Next, Stream functions

$$(u=\partial\psi/\partial y \text{ and } v=-\partial\psi/\partial x)$$

are introduced]

```
>vars:={u(x,y)=diff(psi(x,y),y),v(x,y)=-diff(psi(x,y),x)}: eq:=subs(vars,eq);
```

$$eq := \left(\frac{\partial}{\partial y}\psi(x,y)\right)\left(\frac{\partial^2}{\partial x\,\partial y}\psi(x,y)\right)-\left(\frac{\partial}{\partial x}\psi(x,y)\right)\left(\frac{\partial^2}{\partial y^2}\psi(x,y)\right)-\left(\frac{\partial^3}{\partial y^3}\psi(x,y)\right)$$

[Next, the transformation

$$\psi=\sqrt{x}\,f(\eta), \text{ where } \eta=y/\sqrt{x}$$

is used to obtain the equation for f:]

```
>eq:=changevar(psi(x,y)=x^(1/2)*f(eta(x,y)),eq):
eq1:=(simplify(subs(eta(x,y)=y/x^(1/2),eq))):
eq1:=subs(y=eta*x^(1/2),eq1):eq1:=simplify(eq1*x):eq2:=convert(-eq1,diff);
```

$$eq2 := \frac{1}{2}\left(\frac{d^2}{d\eta^2}f(\eta)\right)f(\eta)+\left(\frac{d^3}{d\eta^3}f(\eta)\right)$$

[Next, the velocity variables, u and v (*i.e.*, derivatives of the stream function), are expressed in terms of f and the similarity variable $\eta$:]

```
>v(eta):=-diff(psi(x,y),x):v(eta):=changevar(psi(x,y)=x^(1/2)*f(eta(x,y)), v(eta)):v(eta):=expand(subs(eta(x,y)=y/x^(1/2),v(eta))):v(eta):= subs(y=eta*x^(1/2),v(eta)):v(eta):=factor(v(eta));
```

$$v(\eta) := -\frac{1}{2}\frac{f(\eta)-D(f)(\eta)\eta}{\sqrt{x}}$$

```
>u(eta):=diff(psi(x,y),y):u(eta):=changevar(psi(x,y)=x^(1/2)*f(eta(x,y)),u(eta)):u(eta):=expand(subs(eta(x,y)=y/x^(1/2),u(eta))): u(eta):=subs(y=eta*x^(1/2),u(eta));
```

$$u(\eta) := D(f)(\eta)$$

[$D(f)(\eta)$ in Maple represents the derivative of f with respect to $\eta$. Next, the boundary conditions are expressed in terms of f:]

```
>bc1:=subs(eta=0,v(eta))=0;
```

$$bc1 := -\frac{1}{2}\frac{f(0)}{\sqrt{x}}=0$$

```
>bc1:=-bc1*2*x^(1/2);
```

$$bc1 := f(0)=0$$

```
>bc2:=subs(eta=0,u(eta))=0;
```

$$bc2 := D(f)(0)=0$$

```
>bc3:=subs(eta=infinity,u(eta))=1;
```

$$bc3 := D(f)(\infty)=1$$

[The length of the domain is taken to be five (to replace infinity). This number is found by trial and error. Increasing the length beyond five does not change the results.]

```
>bc3:=subs(infinity=5,bc3);
```

$$bc3 := D(f)(5) = 1$$

[For this problem, analytical solutions are not possible (although approximate solutions are possible). For this example, numerical solution for the Blasius equation is obtained as:]

```
>sol:=dsolve({eq2,bc1,bc2,bc3},f(eta),type=numeric);
```

sol:= **proc** (x_bvp) ... **end proc**

[The solution is plotted in Figure 2, which shows how the function, f (related to the stream function), varies with the similarity variable, η, from zero to five]

```
>odeplot(sol,[eta,f(eta)],0..5,thickness=3,axes=boxed);
```

[Next, velocity profiles are obtained:]

```
>u(eta):=convert(u(eta),diff);v(eta):=convert(v(eta),diff);
```

$$u(\eta) := \frac{d}{d\eta} f(\eta)$$

$$v(\eta) := -\frac{1}{2}\frac{f(\eta) - \left(\frac{d}{d\eta} f(\eta)\right)\eta}{\sqrt{x}}$$

[Figure 3 shows how the x component of velocity increases from zero, at the wall, and levels off at its main stream value for larger values of η from zero to five]

```
>odeplot(sol,[eta,u(eta)],0..5,thickness=3,axes=boxed
,labels=[eta ,u]);
```

[Since v is a function of x, $v*x^{1/2}$ is plotted. Figure 4 shows the y component of velocity (multiplied by $x^{1/2}$) increases from zero at the wall, and levels off at its main stream value for larger values of η from zero to five]

```
>odeplot(sol,[eta,v(eta)*x^(1/
2)],0..5,thickness=3,axes=boxed,lab
els=[eta,"v*x^(1/2)"]);
```

[The solution at η = 0 is obtained as:]

```
>sol(0);
```

$$\left[\eta=0.,f(\eta)=0.,\frac{d}{d\eta}f(\eta)=0.,\frac{d^2}{d\eta^2}f(\eta)=0.336152378983949952\right]$$




***Figure 2.** Function f as a function of the similarity variable, η.*

[Stress is related to the Reynolds number (re) and the velocity gradient at y = 0:]

```
>S:=re*diff(u(x,y),y);
```

$$S := re\left(\frac{\partial}{\partial y} u(x,y)\right)$$

[The velocity gradient in terms of the stream function is:]

```
>subs(u(x,y)=diff(psi(x,y),y),S);
```

$$re\left(\frac{\partial^2}{\partial y^2}\psi(x,y)\right)$$

[The second derivative of the stream function (d) is expressed in terms of f and η:]

```
>d:=diff(psi(x,y),y$2):d:=changevar(psi(x,y)=x^(1/
2)*f(eta(x,y)), d):d:=expand(subs(eta(x,y)=y/x^(1/2),d)):
d:=subs(y=eta*x^(1/2),d ):d:=convert(d,diff);
```

$$d := \frac{\frac{d^2}{d\eta^2} f(\eta)}{\sqrt{x}}$$

```
>S:=re*d:
```

[The second derivative of f is found from the numerical solution:]

```
>eqd3:=sol(0)[4];
```

$$eqd3 := \frac{d^2}{d\eta^2} f(\eta) = 0.336152378983949952$$

[Hence, the stress-Reynolds number relationship becomes:]

```
>S:=subs(diff(f(eta),`$`(eta,2))=rhs(eqd3),S);
```

$$S := \frac{0.336152378983949952\ re}{\sqrt{x}}$$

## *Example 3*
## Graetz Problem in Rectangular Coordinates

Consider the Graetz problem in rectangular coordinates (to simplify the mathematical complexity involved with cylindrical geometry).[4] The governing equation and initial/boundary conditions are:

$$\left(1-x^2\right)\frac{\partial u}{\partial z} = \frac{\partial^2 u}{\partial x^2}$$
$$u(x,0) = 1 \tag{4}$$
$$u(0,z) = 0 \text{ and } \frac{\partial u}{\partial x}(1,z) = 0$$

For this problem, a similarity transformation cannot be used to reduce the partial differential equation to one ordinary differential equation (boundary value problem in η). To obtain solutions very close to z = 0, Eq. (4) is converted to the new coordinates defined by $\eta = x/\left(2\sqrt{z}\right)$ and $z = z$ (note, some textbooks use $z = z_1$ as the second coordinate, but for simplicity it is left as z in this paper). In the new coordinates, η and z, u is obtained using a perturbation technique by expressing




*Figure 3. The x-component velocity as a function of the similarity variable, η.*




*Figure 4. The y-component velocity as a function of the similarity variable, η.*

u as $u=\sum_{i=0}^{k} Z^k f_i(\eta)$. The boundary conditions for f (in the similarity variable η ) are:

$$f_0(0)=1; f_k(0)=1, k=1,2,3...$$

$$f_0(\infty)=0; f_k(\infty)=0, k=1,2,3... \qquad (5)$$

The steps involved in the similarity transformation method are performed in Maple.

```
>restart:with(student):
>eq:=(1-x^2)*diff(u(x,z),z)-diff(u(x,z),x$2);
```

$$eq := (1-x^2)\left(\frac{\partial}{\partial z}u(x,z)\right)-\left(\frac{\partial^2}{\partial x^2}u(x,z)\right)$$

[First, the governing equation is converted to similarity variables (η and z):]

```
>eq1:=changevar(u(x,z)=U(eta(x,z),z),eq):eq2:=expand(simplify(subs(eta(x,z)=x/2/(z)^(1/2),eq1))):eq2:=expand(eq2*z):eq2:=subs(x=eta*2*(z)^(1/2),eq2):eq2:=convert(eq2,diff):eq2:=expand(-4*eq2);
```

$$eq2 := 2\left(\frac{\partial}{\partial \eta}U(\eta,z)\right)\eta-4z\left(\frac{\partial}{\partial z}U(\eta,z)\right)-8z\eta^3\left(\frac{\partial}{\partial \eta}U(\eta,z)\right)$$
$$+16z^2\eta^2\left(\frac{\partial}{\partial z}U(\eta,z)\right)+\left(\frac{\partial^2}{\partial \eta^2}U(\eta,z)\right)$$

[For illustration, only terms up to $z^2$ are considered in the perturbation series:]

```
>N:=2;vars:={U(eta,z)=sum(z^k*f[k](eta), k=0..N)};
```

$$N:=2$$

$$vars := \{U(\eta,z)=f_0(\eta)+zf_1(\eta)+z^2f_2(\eta)\}$$

[The governing equations for the dependent variables are obtained as:]

```
>eq3:=expand(subs(vars,eq2)):for i from 0 to 2 do Eq[i]:=coeff(eq3,z,i);od;
```

$$Eq_0 := 2\eta\left(\frac{d}{d\eta}f_0(\eta)\right)+\left(\frac{d^2}{d\eta^2}f_0(\eta)\right)$$

$$Eq_1 := 2\eta\left(\frac{d}{d\eta}f_1(\eta)\right)-4f_1(\eta)-8\eta^3\left(\frac{d}{d\eta}f_0(\eta)\right)+\left(\frac{d^2}{d\eta^2}f_1(\eta)\right)$$

$$Eq_2 := 2\eta\left(\frac{d}{d\eta}f_2(\eta)\right)-8f_2(\eta)-8\eta^3\left(\frac{d}{d\eta}f_1(\eta)\right)$$
$$+16\eta^2 f_1(\eta)+\left(\frac{d^2}{d\eta^2}f_2(\eta)\right)$$

[The first three terms are obtained by solving these differential equations with the given boundary conditions (note that the boundary condition at x = 1 is solved approximately as U = 0 at η = ∞:]

```
>sol[0]:=dsolve({Eq[0],f[0](0)=0,f[0](infinity)=1});assign(sol[0] ):
```

$$sol_0 := f_0(\eta) = erf(\eta)$$

```
>sol[1]:=dsolve({Eq[1]});
```

$$sol_1 := \left\{ f_1(\eta)=(1+2\eta^2)\_C2 \right.$$
$$\left. +(1+2\eta^2)\int \frac{e^{(-\eta^2)}}{(1+2\eta^2)^2}d\eta\_C1+\frac{1}{3}\frac{(-3\eta-4\eta^3)e^{(-\eta^2)}}{\sqrt{\pi}} \right\}$$

[The constants have to be zero to satisfy the boundary conditions:]

```
>assign(sol[1]):_C1:=0:_C2:=0:f[1](eta):=eval(f[1](eta));
```

$$f_1(\eta) := \frac{1}{3}\frac{(-3\eta-4\eta^3)e^{(-\eta^2)}}{\sqrt{\pi}}$$

[Similarly, $f_2$ is obtained:]

```
>sol[2]:=dsolve(Eq[2]):assign(sol[2]):_C3:=0:_C4:=0:
f[2](eta):=eval(f[2](eta));
```

$$f_2(\eta) := \frac{1}{180}\frac{\left(-285\eta - 570\eta^3 - 384\eta^5 - 160\eta^7\right)e^{(-\eta^2)}}{\sqrt{\pi}}$$

[Once the functions (the f's) are obtained, the Sherwood number can be obtained:[4]]

```
>u:=subs(vars,U(eta,z)):u:=subs(eta=x/2/sqrt(z),u);
```

$$u := \mathrm{erf}\left(\frac{x}{2\sqrt{z}}\right) + \frac{1}{3}\frac{\left(-\frac{3x}{2\sqrt{z}} - \frac{x^3}{2z^{(3/2)}}\right)e^{\left(-\frac{x^2}{4z}\right)}}{\sqrt{\pi}} + \frac{1}{180}\frac{z^2\left(-\frac{285x}{2\sqrt{z}} - \frac{285x^3}{4z^{(3/2)}} - \frac{12x^5}{z^{(5/2)}} - \frac{5x^7}{4z^{(7/2)}}\right)e^{\left(-\frac{x^2}{4z}\right)}}{\sqrt{\pi}}$$

[The dimensionless temperature distribution is plotted in Figure 5, which shows that temperature increases from the center of the slab to the surface value along the x-coordinate. The increase in temperature is more rapid at the entrance and temperature increases are more gradual for higher values of z from 0 to 0.05, the distance along the flow.]

```
>plot3d(u,x=1..0,z=0.05..0,axes=boxed,labels=[x,z,
"u"],orientatio n=[120,60]);
```

## SUMMARY

This paper illustrates that mathematical methods for nontrivial problems in chemical engineering can be taught efficiently in a class using computers and user-friendly software.

The similarity solution approach is a very powerful technique for obtaining closed-form solutions for problems in heat, mass, momentum transfer, and other disciplines in chemical engineering. A traditional approach to teaching this technique would involve complicated variable transformations and integrals done by hand. In this paper, it was shown how an analytical technique could be facilitated using computers and software. While Maple has been used in this paper, Mathematica, MATLAB, REDUCE, or other symbolic software packages can be used to obtain similar results. In addition to teaching numerical simulation, computers and software packages can be used to teach traditional mathematical methods for a wide variety of problems. Mathematical methods, such as separation of variables, Laplace transform, perturbation, conformal mapping, Green's function, analytical method of lines, and series solutions for nonlinear problems (multiple steady states) can be facilitated using Maple. Readers can contact the author for further details or copies of related Maple programs. Some of these methods are illustrated in a book to be published in the future.[9]

## REFERENCES


1. <http://www.maplesoft.com>
2. <http://www.wolfram.com>
3. <http://www.mathworks.com>
4. <http://www.reduce-algebra.com>
5. Carslaw, H.S., and J.C. Jaeger, *Conduction of Heat in Solids*, Oxford University Press, London (1973)
6. Crank, J., *Mathematics of Diffusion*, Oxford University Press, New York (1975)
7. Slattery, J., *Advanced Transport Phenomena*, Cambridge University Press, New York (1999)
8. Villedsen, J., and M.L. Michelsen, *Solution of Differential Equation Models by Polynomial Approximation*, Prentice-Hall, Englewood Cliffs, NJ (1978)
9. White, R.E., and V.R. Subramanian, *Computational Methods in Chemical Engineering with Maple Applications*, Springer-Verlag (to be submitted in 2006). ❐





***Figure 5.*** *Dimensionless temperature distribution in rectangular coordinates, governed by the Graetz equation.*